


```
Out[6]: %6 =  
[x^2 - 2*x + 2 1]  
  
[x^2 + 2*x + 2 1]
```

```
In [7]: intnum(x=-oo,oo,1/(1+x^2))
```

```
Out[7]: %7 = 3.1415926535897932384626433832795028842
```

```
In [8]: sin(x)
```

```
Out[8]: %8 = x - 1/6*x^3 + 1/120*x^5 - 1/5040*x^7 + 1/362880*x^9 - 1/39916800*x^11 + 1/62270  
20800*x^13 - 1/1307674368000*x^15 + 0(x^17)
```

```
In [9]: sin(x+0(x^20))
```

```
Out[9]: %9 = x - 1/6*x^3 + 1/120*x^5 - 1/5040*x^7 + 1/362880*x^9 - 1/39916800*x^11 + 1/62270  
20800*x^13 - 1/1307674368000*x^15 + 1/355687428096000*x^17 - 1/121645100408832000*x^  
19 + 0(x^20)
```

```
In [10]: s=sqrt(2+0(7^10))
```

```
Out[10]: %10 = 3 + 7 + 2*7^2 + 6*7^3 + 7^4 + 2*7^5 + 7^6 + 2*7^7 + 4*7^8 + 6*7^9 + 0(7^10)
```

```
In [11]: s^2
```

```
Out[11]: %11 = 2 + 0(7^10)
```

Théorie de Galois élémentaire

```
In [12]: P=x^4+1
```

```
Out[12]: %12 = x^4 + 1
```

```
In [13]: G=galoisinit(P);
```

```
In [14]: [d,t]=galoisidentify(G)
```

```
Out[14]: %14 = [4, 2]
```

```
In [15]: galoisgetname(d,t)
```

```
Out[15]: %15 = "C2 x C2"
```

```
In [16]: S=galoissubgroups(G);
```

```
In [17]: F=galoissubfields(G,1)
```

```
Out[17]: %17 = [x, x^2 - 2, x^2 + 4, x^2 + 2, x^4 + 1]
```

```
In [18]: galoisfixedfield(G,S[2],2)
```

```
Out[18]: %18 = [y^2 - 2, Mod(-x^3 + x, x^4 + 1), [x^2 - y*x + 1, x^2 + y*x + 1]]
```

Nous avons la factorisation $x^4 + 1 = (x^2 - \sqrt{2}x + 1)(x^2 + \sqrt{2}x + 1)$

```
In [19]: Mod(-x^3 + x, x^4 + 1)^2
```

```
Out[19]: %19 = Mod(2, x^4 + 1)
```

```
In [20]: galoisfixedfield(G,S[3],2)
```

```
Out[20]: %20 = [y^2 + 4, Mod(2*x^2, x^4 + 1), [x^2 - 1/2*y, x^2 + 1/2*y]]
```

Nous avons la factorisation $x^4 + 1 = (x^2 - \sqrt{-1})(x^2 + \sqrt{-1})$

```
In [21]: galoisfixedfield(G,S[4],2)
```

```
Out[21]: %21 = [y^2 + 2, Mod(x^3 + x, x^4 + 1), [x^2 - y*x - 1, x^2 + y*x - 1]]
```

Nous avons la factorisation $x^4 + 1 = (x^2 - \sqrt{-2}x - 1)(x^2 + \sqrt{-2}x - 1)$

Théorie algébrique des nombres

```
In [22]: K=bnfinit(alpha^2-79);
```

Initialise $K = \mathbb{Q}(\alpha)$ avec $\alpha = \sqrt{79}$

```
In [23]: K.zk
```

```
Out[23]: %23 = [1, alpha]
```

$\mathcal{O}_K = \mathbb{Z} + \alpha\mathbb{Z} = \mathbb{Z}[\alpha]$

```
In [24]: K.disc
```

```
Out[24]: %24 = 316
```

Le discriminant de K vaut 316

```
In [25]: K.sign
```

```
Out[25]: %25 = [2, 0]
```

K est totalement réel.

```
In [26]: K.cyc
```

```
Out[26]: %26 = [3]
```

Le groupe de classe est $\mathbb{Z}/3\mathbb{Z}$.

```
In [27]: K.fu
```

```
Out[27]: %27 = [Mod(9*alpha + 80, alpha^2 - 79)]
```

L'unité fondamentale est $80 - 9\alpha$.

```
In [28]: 80^2-79*9^2
```

```
Out[28]: %28 = 1
```

```
In [29]: K.reg
```

```
Out[29]: %29 = 5.0751347504448098597876951184786988027
```

Le régulateur vaut $\simeq 5.075134750444809$

```
In [30]: log(K.fu[1])
```

```
Out[30]: %30 = [-5.0751347504448098597876951184786988598, 5.075134750444809859787695118478698  
8038]~
```

```
In [31]: id2 = idealprimedec(K,2);
```

id2 est la décomposition de 2 dans le corps K .

```
In [32]: g2=#id2
```

```
Out[32]: %32 = 1
```

```
In [33]: id2[1].e
```

```
Out[33]: %33 = 2
```

```
In [34]: id2[1].f
```

```
Out[34]: %34 = 1
```

$$2\mathcal{O}_K = \mathfrak{p}_2^2$$

```
In [35]: bnfisprincipal(K,id2[1])
```

```
Out[35]: %35 = [[0]~, [-9, -1]~]
```

$$\mathfrak{p}_2 = (9 + \alpha)$$

```
In [36]: 9^2-79*1^2
```

```
Out[36]: %36 = 2
```

```
In [37]: id3 = idealprimedec(K,3);
```

```
In [38]: g3=#id3
```

```
Out[38]: %38 = 2
```

```
In [39]: id3[1].e
```

```
Out[39]: %39 = 1
```

```
In [40]: id3[1].f
```

```
Out[40]: %40 = 1
```

$$3\mathcal{O}_K = \mathfrak{p}_3\mathfrak{p}'_3$$

```
In [41]: bnfisprincipal(K,id3[1])
```

```
Out[41]: %41 = [[1]~, [1, 0]~]
```

\mathfrak{p}_3 n'est pas principal

```
In [42]: p3 = idealpow(K, id3[1], 3);
```

$$p3 = \mathfrak{p}_3^3$$

```
In [43]: bnfisprincipal(K,p3)
```

```
Out[43]: %43 = [[0]~, [-17, 2]~]
```

\mathfrak{p}_3^3 est principal et engendré par $2\alpha - 17$.

```
In [44]: 17^2-79*2^2
```

```
Out[44]: %44 = -27
```

```
In [45]: id5 = idealprimedec(K,5);
```

```
In [46]: g5 = #id5
```

```
Out[46]: %46 = 2
```

```
In [47]: id5[1].e
```

```
Out[47]: %47 = 1
```

```
In [48]: id5[1].f
```

```
Out[48]: %48 = 1
```

$$5\mathcal{O}_K = \mathfrak{p}_5\mathfrak{p}'_5$$

```
In [49]: bnfisprincipal(K,id5[1])
```

```
Out[49]: %49 = [[1]~, [-8/3, 1/3]~]
```

\mathfrak{p}_5 est dans la même classe d'idéal que \mathfrak{p}_3 .

```
In [50]: p35 = idealmul(K, id3[1], id5[2]);
```

$$\mathfrak{p}_{35} = \mathfrak{p}_3\mathfrak{p}'_5$$

```
In [51]: bnfisprincipal(K,p35)
```

```
Out[51]: %51 = [[0]~, [8, 1]~]
```

$$\mathfrak{p}_3\mathfrak{p}'_5 = (-8 - \alpha)$$

```
In [52]: 8^2-79*1^2
```

```
Out[52]: %52 = -15
```

```
In [53]: bnfisintnorm(K,1049)
```

```
Out[53]: %53 = [13*alpha + 120, 40*alpha + 357]
```

```
In [54]: 120^2-79*13^2
```

```
Out[54]: %54 = 1049
```

Fonctions L

```
In [55]: Z=lfuninit(K,[1/2,1/2,100],1);
```

$Z = \zeta_K(s) = \prod_p \frac{1}{1 - \text{Norm}(p)^{-s}}$ la fonction zêta de Dedekind de K

```
In [56]: lfun(Z,0,0)
```

```
Out[56]: %56 = 0
```

```
In [57]: lfun(Z,0,1)
```

```
Out[57]: %57 = -7.6127021256672147896815426777180482058
```

```
In [58]: -K.no*K.reg/2
```

```
Out[58]: %58 = -7.6127021256672147896815426777180482040
```

L'égalité vient de la formule du nombre de classe de Dirichlet.

```
In [59]: lfun(Z,1)
```


```
Out[59]: %59 = 1.7129918109884013874307672625564043042*x^-1 + 0(x^0)
```

```
In [60]: 2^K.r1*K.no*K.reg/sqrt(K.disc)/2
```


```
Out[60]: %60 = 1.7129918109884013874307672625564043038
```

L'égalité vient aussi de la formule du nombre de classe de Dirichlet.


```
In [61]: ploth(x=0,10,lfun(Z,.3+I*x),"Complex");
```



```
In [62]: ploth(x=0,10,lfun(Z,.4+I*x),"Complex");
```



```
In [63]: plot(x=0,10,lfun(Z, .5+I*x), "Complex");
```



L'hypothèse de Riemann pour ζ_K est que les zéros sont de partie réel $1/2$.

Théorie du corps de classes

```
In [64]: B=bnrclassfield(K)
```

```
Out[64]: %64 = [x^3 - 21*x - 4*alpha]
```

```
In [65]: L=rnfinit(K,B[1]);
```

```
In [66]: L.disc
```

Out[66]: %66 = [1, 1]

L'idéal discriminant relatif vaut 1.

```
In [67]: rnfisabelian(K,B[1])
```

Out[67]: %67 = 1

L'extension est abélienne.

```
In [68]: p1049=idealprimedec(K,1049);
```

```
In [69]: #p1049
```

Out[69]: %69 = 2

1049 est totalement décomposé dans K

```
In [70]: #rnfidealprimedec(L,p1049[1])
```

Out[70]: %70 = 3

\mathfrak{p}_{1049} est totalement décomposé dans L et donc principal.

```
In [71]: bnfisprincipal(K,p1049[1])
```

Out[71]: %71 = [[0]~, [120, -13]~]

Équations diophantiennes

Résoudre $x^2 - 6712y^2 = 1$ avec x, y entiers.

```
In [72]: bnf=bnfinit(alpha^2-6712);bnf.fu
```

Out[72]: %72 = [Mod(119343246469604298*alpha + 9777409878270240143, alpha^2 - 6712)]

```
In [73]: 9777409878270240143^2-6712*119343246469604298^2
```

Out[73]: %73 = 1

Résoudre $y^2 = x^3 - 157^2x$, avec x, y rationels.

```
In [74]: E=ellinit([-157^2,0]); P=ellheegner(E)
```

Out[74]: %74 = [69648970982596494254458225/166136231668185267540804, 538962435089604615078004
307258785218335/67716816556077455999228495435742408]

```
In [75]: P[1]^3-157^2*P[1] - P[2]^2
```

Out[75]: %75 = 0

Résoudre $x^5 - 2y^5 = 3$ avec x, y entiers.

```
In [76]: thue(x^5-2,3)
```

Out[76]: %76 = [[1, -1]]

Il y a seulement la solution triviale $x = 1, y = -1$

```
In [77]: bnf=bnfinit(alpha^3-2);
```

Résoudre $x^3 + 2y^3 + 4z^3 - 6xyz = 10009$

```
In [78]: bnfisintnorm(bnf,10009)
```

```
Out[78]: %78 = [-6*alpha^2 + 13*alpha + 11, 13*alpha^2 + 10*alpha + 1, -6*alpha^2 + 15*alpha + 7]
```

```
In [79]: 11^3+2*13^3-4*6^3-6*11*13*-6
```

```
Out[79]: %79 = 10009
```

Interpolation

```
In [80]: exp(Pi*sqrt(163))
```

```
Out[80]: %80 = 262537412640768743.9999999999925007259
```

```
In [ ]:
```

```
In [81]: 744-ellj((1+sqrt(-163))/2)
```

```
Out[81]: %81 = 262537412640768744.00000000000000000000
```

```
In [82]: ellj((1+sqrt(-31))/2)
```

```
Out[82]: %82 = -39492793.911556244143880327445303424864
```

```
In [83]: algdep(%,3)
```

```
Out[83]: %83 = 214732*x^3 + 8480366314537*x^2 - 12151115349442*x - 13611811239453
```

```
In [84]: localprec(100); z=ellj((1+sqrt(-31))/2)
```

```
Out[84]: %84 = -39492793.911556244143880327445303424866
```

```
In [85]: algdep(z,3)
```

```
Out[85]: %85 = x^3 + 39491307*x^2 - 58682638134*x + 1566028350940383
```

```
In [86]: polclass(-31)
```

```
Out[86]: %86 = x^3 + 39491307*x^2 - 58682638134*x + 1566028350940383
```

```
In [87]: z=sumpos(n=1, sumpos(m=n+1, 1/(n^2*m^5)))
```

```
Out[87]: %87 = 0.038575124342753255505925464372562108149
```

```
In [88]: linddep([z,zeta(7),zeta(5)*zeta(2),zeta(3)*zeta(4)])
```

```
Out[88]: %88 = [1, 11, -5, -2]~
```

```
In [89]: 5*zeta(5)*zeta(2)+2*zeta(3)*zeta(4)-11*zeta(7)
```

```
Out[89]: %89 = 0.038575124342753255505925464372995570013
```

```
In [90]: 1/4+0(5^20)
```

```
Out[90]: %90 = 4 + 3*5 + 3*5^2 + 3*5^3 + 3*5^4 + 3*5^5 + 3*5^6 + 3*5^7 + 3*5^8 + 3*5^9 + 3*5^10 + 3*5^11 + 3*5^12 + 3*5^13 + 3*5^14 + 3*5^15 + 3*5^16 + 3*5^17 + 3*5^18 + 3*5^19 + 0(5^20)
```

```
In [91]: s=gamma(1/4+0(5^20))
```

```
Out[91]: %91 = 1 + 4*5 + 3*5^4 + 5^6 + 5^7 + 4*5^9 + 5^10 + 2*5^12 + 5^13 + 2*5^14 + 5^15 + 3*5^16 + 2*5^18 + 3*5^19 + 0(5^20)
```

```
In [92]: algdep(s,4)
```

```
Out[92]: %92 = x^4 + 4*x^2 + 5
```

```
In [93]: s^4+4*s^2+5
```


```
Out[93]: %93 = 0(5^20)
```

Formule explicite de Riemann

```
In [94]: default(graphcolors,[4,2,5]);
```

```
In [95]: Cpsi(n)=log(lcm([1..floor(n)]));
```

```
In [96]: ploth(x=1.1,30,[Cpsi(x),x]);
```




```
In [97]: Z=lfunzeros(1,1000);
```

```
In [98]: F(x,s=#Z)=my(lx=I*log(x));x-2*sqrt(x)*real(sum(i=1,s,exp(lx*Z[i])/(1/2+I*Z[i]))) - log(
```

Fonction ψ de Chebyshev

```
In [110... ploth(x=1.1,30,[F(x,5),Cpsi(x),x])
```



Out[110... %106 = [1.1000000000000000888, 30.000000000000000000, -0.16847039935043686687, 30.000000000000000000]

In []: